

INTERPOLATION ALS DYNAMISCHER PROZEß

W.-W. Scheuermann Hamburg Okt. 85

In vielen Anwendungen steht man vor dem Problem, zwischen gegebenen Stützpunkten interpolieren zu müssen. Die Aufgabe der Interpolation ist dabei, eine Kurve mathematisch zu beschreiben oder numerisch zu berechnen, welche neben der wesentlichen Eigenschaft, durch die Stützpunkte zu verlaufen, einige weitere Kriterien erfüllen soll:

- die interpolierte Kurve soll "glatt", ohne scharfe Knicke sein, d.h. mindestens in der ersten Ableitung stetig sein.
- die Kurve soll "unverschnörkelt" sein, d.h. keine Überschwinger aufweisen und möglichst wenig vom kürzesten Weg zwischen den Stützpunkten abweichen.

Lösungen des Interpolationsproblems bestehen in mathematisch definierten Polynomen oder Splines und numerisch berechneten Kurven oder geometrisch konstruierten Kurvenstücken.

Der im folgenden beschriebene Ansatz basiert auf einer Diskussion mit H. Stoltenberg und resultiert aus der Untersuchung der Frage, wie ein menschlicher Zeichner vorgeht, wenn er eine gegebene Punktmenge mit einer glatten Kurve verbindet.

Zeichnen ist ein dynamischer Prozeß, der ständiger Kontrolle und Regelung unterliegt. Es ist allerdings kein physikalischer Prozeß in dem Sinne, daß "Kräfte" von den Stützpunkten ausgingen, die den Zeichenstift beeinflussen. Der Stift kann mit konstanter Bahngeschwindigkeit geführt werden und lediglich sein Kurs wird geregelt.

Betrachten wir das einfachste, typische Interpolationsproblem: Drei Punkte sollen verbunden werden:




Abb.1 einfaches Interpolationsproblem

Im Vergleich mit der direkten Verbindung der Punkte zeigt sich eine charakteristische Abweichung zur konvexen Seite hin: Offenbar bewirkt eine nach dem nächsten Punkt vorzunehmende Kursänderung eine Abweichung von der direkten Richtung auf diesen Punkt zu. Die Abweichung ist dabei umso größer, je größer die nötige Kursänderung ist. Die Abweichung erfolgt nach derjenigen Seite, die es erlaubt, den nächsten Punkt so anzusteuern, daß die verbleibende Kursänderung verkleinert wird. Die Bahn des Zeichenstiftes wird also im wesentlichen bestimmt durch die Lage der nächsten beiden Punkte und dem momentanen Kurs der Linie.

Die Implementierung dieser heuristischen Prinzipien führt zu einem Algorithmus, der in befriedigender Weise das Interpolationsproblem löst und auch zu ästhetisch überzeugenden Ergebnissen führt. Umgekehrt könnte deshalb der Algorithmus auch für die theoretische Erklärung der Koordinationsprozesse beim Zeichnen von Kurven dienen: So oder doch so ähnlich gehen auch Menschen beim zeichnerischen Interpolieren vor.

Ich verwende folgenden Regelungsmechanismus:

Der Zeichenstift bewegt sich mit konstanter Schrittweite r in jedem Rechenzyklus n in eine bestimmte Richtung β_n . Diese Richtung β_n ist bestimmt von der Richtung im vorhergehenden Rechenzyklus β_{n-1} und einem Korrekturwinkel δ :

$$\beta_n = \beta_{n-1} + \delta$$

δ wiederum hängt von drei Größen ab: von der momentanen Distanz d des Zeichenstiftes zum nächsten Stützpunkt, der Richtungsdivergenz ϑ_1 zwischen momentaner Bewegungsrichtung β_n und Richtung auf den Stützpunkt hin sowie der Winkeldifferenz ϑ_2 zwischen β_n und der Richtung zwischen dem nächsten und dem übernächsten Stützpunkt:

$$\delta = \vartheta_1 e^{-u\sqrt{d}} - v\vartheta_2$$

Die Regelung erfolgt zweistufig, wobei der erste Term der Korrekturfunktion abhängig vom nächsten Punkt die erste Stufe, der zweite Term, abhängig vom übernächsten Punkt die zweite Stufe darstellt. Mit den Parametern u und v können die Stufen der Korrekturfunktion abgestimmt werden.




Abb.2 Der dynamische Prozeß des Zeichnens

Die Funktion des Algorithmus und der Einfluß der Parameter sei an folgenden Beispielen demonstriert.




Abb.3 ZU interpolierende Punktmenge

Volle Regelung 1. Stufe, keine Regelung 2. Stufe:
Man beachte, daß der Verlauf zwischen den Punkten 8,9 und 10
konvex ist.

ANFANGSKURS? 130
U=0 V=0




Abb. 4

ANFANGSKURS? 320
U=0 V=0




Abb. 5

Regelung 1. Stufe, keine Regelung 2. Stufe:

Die Darstellungen zeigen die komplementäre Wirkungsweisen, wenn man sie mit dem Einfluß der 2. Stufe vergleicht (Vgl. Abb.6 mit Abb.9 und Abb.7 mit Abb.8)

ANFANGSKURS? 130

U=.415 V=0




Abb.6

ANFANGSKURS? 320

U=.4 V=0




Abb.7

Volle Regelung 1. Stufe, Regelung 2. Stufe:
In der Abweichung von der direkten Verbindung zeigt sich
die Wirkung der Regelung der 2. Stufe.

ANFANGSKURS? 130

U=0 V=.237




Abb. 8

ANFANGSKURS? 320

U=0 V=.236





Abb. 9


Abgestimmte Regelung 1. und 2. Stufe:



ANFANGSKURS? 130

U=.3525 V=.025

Abb.10



ANFANGSKURS? 320

U=.3525 V=.025

Abb.11

U=0.1925 V=0.086

Abb.12




Abb.13 Linienzüge


So rum siehts besser aus!

ULIST

```
1000 REM -----
1010 REM -----KURVE.5-----
1020 REM -----4.10.85-----
1030 REM -----W.-W.SCHEUERMANN-----
1040 REM -----
1050 REM DAS PROGRAMM KURVE.5 VERBINDET STUETZPUNKTE MIT EINER STETIGEN KURVE.
1060 REM KEINE DER GAENGIGEN INTERPOLATIONSMETHODEN WIRD VERWENDET, SONDERN DIE
1070 REM TAETIGKEIT EINES ZEICHNERS VOR DERSELBEN AUFGABE WIRD SIMULIERT. DER
1080 REM "ZEICHENSTIFT" BEWEGT SICH MIT GLEICHFOERMIGER GESCHWINDIGKEIT, WOBEI
1090 REM ALLERDINGS SEINE RICHTUNG EINER STAENDIGEN KONTROLLE UND REGELUNG UNTER-
1100 REM LIEGT. DER REGELMECHANISMUS IST EMPIRISCH, SEINE PRINZIPIEN HEURISTISCH.
1110 REM FOLGENDE KRITERIEN SPIELEN DABEI FUER DIE RICHTUNGSAENDERUNG EINE
1120 REM ROLLE:
1130 REM - DIE DIFFERENZ ZWISCHEN MOMENTANER BEWEGUNGSRICHTUNG UND PEILUNG DES
1140 REM NAECHSTEN PUNKTES
1150 REM - DER ABSTAND ZUM NAECHSTEN PUNKT
1160 REM - DIE KURSDIFFERENZ ZWISCHEN MOMENTANER BEWEGUNGSRICHTUNG UND DER
1170 REM RICHTUNG ZWISCHEN DEM NAECHSTEN UND DEM DARAUFFOLGENDEN PUNKT (VORAUS-
1180 REM SCHAU)
1190 REM ZWEI PARAMETER (U,V) ERLAUBEN DIE ABSTIMMUNG DER EINFLUESSE.
1200 REM -----
1210 :
1220 REM -----VEREINBARUNGSTEIL-----
1230 HOME : REM BILDSCHIRM LOESCHEN
1240 PI = ATN (1) * 4: REM KREISZAHL
1250 RAD = PI / 180: REM UMRECHNUNGSFAKTOR GRAD->RAD
1260 R = 1: REM SCHRITTWEITE DES ZEICHENSTIFTS VON ZYKLUS ZU ZYKLUS
1270 NN = 12: REM ANZAHL DER STUETZPUNKTE
1280 INPUT "ANFANGSKURS? ";BETA
1290 BETA = BETA * RAD
1295 DIM X(20),Y(20): REM ARRAYS DIMENSIONIEREN
1300 REM -----PUNKTE EINLESEN-----
1301 REM ES EMPFIEHLT SICH, EINEN PUNKT MEHR ALS DARGESTELLT ZU DEFINIEREN, UM
1302 REM EINEN DEFINIERTEN AUSLAUFKURS ZU ERHALTEN. BEI IN SICH GESCHLOSSENEN
1303 REM KURVEN MUSS DESHALB DER VORLETZTE PUNKT MIT DEM ERSTEN IDENTISCH SEIN
1304 REM UND DER LETZTE MIT DEM ZWEITEN.
1310 FOR I = 1 TO NN + 1
1320 READ X(I),Y(I)
1323 NEXT I
1324 REM -----DATEN DER STUETZPUNKTE-----
1325 REM DATA X-WERT,Y-WERT
1326 DATA 30,40
1327 DATA 80,40
1328 DATA 110,60
1329 DATA 130,50
1330 DATA 140,40
1331 DATA 150,20
1332 DATA 170,50
1333 DATA 160,80
1334 DATA 110,100
1335 DATA 70,110
1336 DATA 30,90
1337 DATA 30,40
1338 DATA 80,40
1340 REM -----PLOT VORBEREITEN-----
1350 HGR2 : HCOLOR= 3: REM GRAFIK-PAGE UND FARBE WAELLEN
```

```
1360 FOR I = 1 TO NN
1370 HPLLOT X(I),Y(I)
1380 NEXT I
1390 :
1400 REM -----BEGINN-----
1410 K = 1: REM STUETZPUNKTZAehler INITIALISIEREN
1420 X = X(K): REM ZEICHENSTIFT AUF ERSTEN PUNKT SETZEN
1430 Y = Y(K)
1440 REM -----SCHLEIFENANFANG-----
1450 XA = X(K + 1) - X: REM PEILUNG ZUM NAECHSTEN PUNKT BESTIMMEN
1460 YA = Y(K + 1) - Y: REM DIFFERENZVEKTOR BERECHNEN
1470 GOSUB 2100: REM WINKELBESTIMMUNG
1480 A1 = ALFA: REM 1.RICHTUNG
1490 XA = X(K + 2) - X(K + 1): REM 2.RICHTUNGSVEKTOR BERECHNEN
1500 YA = Y(K + 2) - Y(K + 1)
1510 GOSUB 2100: REM WINKELBESTIMMUNG
1520 A2 = ALFA: REM 2.RICHTUNG
1530 REM -----PARAMETER-----
1540 U = PDL (1) / 400: REM EXTERNE ANALOGEINGABE (PADDLES)
1550 V = PDL (0) / 1000
1560 REM -----BERECHNUNG DER REGELGROESSE-----
1570 DIST = SQR ((X - X(K + 1)) ^ 2 + (Y - Y(K + 1)) ^ 2): REM ABSTAND ZUM NAECHSTEN PU
NKT
1580 D1 = A1 - BETA: REM ROHDIFFERENZ 1
1590 D2 = A2 - BETA: REM ROHDIFFERENZ 2
1600 R1 = D1 + SGN ( SIN (D1)) * 2 * PI * ( ABS (D1) > PI): REM DIFFERENZ 1 KORRIGIERT,
WENN D1>180 (ABS(D1)>PI) UND MIT DEM RICHTIGEN VORZEICHEN VERSEHEN (SGN(SIN(D1)))
1620 R2 = D2 + SGN ( SIN (D2)) * 2 * PI * ( ABS (D2) > PI): REM DIFFERENZ 2 KORRIGIERT
ANALOG R1
1630 D = R1 * EXP ( - SQR (DIST) * U) - V * R2: REM WINKELKORREKTUR, BERECHNET AUS DEM
1640 REM ENTFERNUNGSABHAENGIGEN R1 (JE NAEHER, DESTO STAERKER) UND DEM ENTGEGENGE-
1650 REM SETZT WIRKENDEN R2 (WIRKT VORKRUEMMEND FUER DIE FOLGENDE RICHTUNG)
1660 REM DIE DISTANZABHAENGIGKEIT UND DIE VORKRUEMMUNG SIND PARAMETERKONTROLLIERT
1670 BETA = BETA + D: REM NEUE RICHTUNG DES ZEICHENSTIFTES
1680 IF BETA < 0 THEN BETA = BETA + 2 * PI
1685 IF BETA > 2 * PI THEN BETA = BETA - 2 * PI: REM VOLLKREISIG
1690 REM -----ZEICHNEN-----
1700 X = X + R * SIN (BETA): REM STIFT IN DIE BERECHNETE RICHTUNG UM EINEN SCHRITT WEIT
ER
1710 Y = Y + R * COS (BETA)
1720 HPLLOT X,Y: REM ZEICHNEN
1730 REM -----SCHLEIFENENDE-----
1740 IF DIST > 1.5 GOTO 1450: REM FANGKREIS UM NAECHSTEN PUNKT
1750 K = K + 1: REM NAECHSTER PUNKT
1760 IF K = NN THEN PRINT : PRINT "U="U" V="V: PRINT : PRINT : END : REM ABRUCHBEDIN
GUNG: LETZTER PUNKT ERREICHT
1770 GOTO 1450: REM SCHLEIFE FORTSETZEN
1780 :
1790 :
2000 REM -----
2010 REM -----SUBROUTINE: BERECHNUNG DER PEILUNG-----
2020 REM -----
2100 IF XA = 0 AND YA = 0 THEN RETURN : REM IN DIESEM FALL: ALTEN KURS BEIBEHALTEN
2107 ALFA = ATN (XA / (YA + 1E - 30)): REM WINKELBERECHNUNG MIT VERMEIDUNG DER DIVISION
DURCH NULL
2120 ALFA = ALFA + PI * ((XA > 0 AND YA < 0) OR (XA < = 0 AND YA < 0)) + PI * 2 * ((XA <
0 AND YA > 0) OR (XA < 0 AND YA = 0))
2130 REM ALFA VOLLKREISIG. VORSICHT MIT DEN LOGISCHEN BEDINGUNGEN: NOCH KOENNEN
2140 REM FEHLER VORHANDEN SEIN (VOR ALLEM GLEICHHEITSBEDINGUNGEN!)
2150 RETURN : REM ENDE SUBROUTINE
2160 REM -----
```

Ü



künstliche Kreativität

.042

Ü



Ü?U, V
.385

.012

ÜRÜN